
ABSTRACT 

 Current standards for reduction of observed gravity to a modeled Bouguer 

anomaly largely are unregulated and vary among geophysical textbooks, commercial 

software programs, and academic research spreadsheets available for download from the 

internet.  Using new standards established by the U.S. Geological Survey and the North 

American Gravity Database Committee, we developed a spreadsheet for reduction of raw 

data to the Bouguer anomaly and, with the use of terrain correction, the Complete 

Bouguer anomaly.  The reduction is based on ellipsoidal height.  Included in the data 

reduction package is a U.S. Geodetic Survey transformation program that enables 

restoration of orthometric height readings (typical of legacy data) to ellipsoidal height.  

The spreadsheet is available for free download at Geological Society of America data 

repository and geongrid.org.  We view the spreadsheet as particularly useful for field data 

reduction and modeling where internet access is limited or unavailable. 

INTRODUCTION 

With the use of the Global Positioning System (GPS) for surveying station 

locations and altitudes, availability of digital terrain models, and enhanced computational 

capability, gravity modeling is a cost-effective tool in subsurface analysis ranging from 

basin to continental scale studies.  Existing gravity data for North America are archived 

and readily accessible via the internet at the Pan-American Center for Earth and 

Environmental Studies website (http://paces.geo.utep.edu/).  The North America gravity 

database provides principal facts and Free Air and Bouguer anomalies calculated by a 

FORTRAN algorithm based on preferred correction and anomaly equations established 

by the Standards/Format Working Group of the North American Gravity Database 

Committee (Hinze et al., 2002).    

To facilitate adoption of the standards established by the North American Gravity 

Database Committee (Hinze et al., 2002) by the research community and to provide an 

easy to use, portable gravity correction and anomaly computation platform, we developed 

a gravity spreadsheet.  The spreadsheet is based on Microsoft Excel, which is a common 

software application used by government agencies, research institutions, and private 

companies.  The equations used in the spreadsheet are derived from the FORTRAN code 

written by Mike Webring of the U.S. Geological Survey and are the same as those used 

by the GeoNet Server at the Pan-American Center for Earth and Environmental Studies 

(PACES, 2006).    



 

STANDARDIZED GRAVITY REDUCTION 

The equations described in this section are used in the gravity spreadsheet and 

conform to the new gravity standards set by the U.S. Geological Survey (Hildebrand, 

2002) and the Standards/Format Working Group of the North American Gravity 

Database Committee (Hinze et al., 2003).  

 

Ellipsoid Theoretical Gravity 

The ellipsoid theoretical gravity calculation uses the Somigliana closed-form 

formula based on the 1980 Geodetic Reference System (GRS80) to predict the gravity at 

any latitude φ north or south (Hildebrand, 2002).   

where values for the GRS80 reference ellipsoid are: ge = 978032.67715 mGals, k = 

0.001931851353 is a dimensionless coefficient, e2 = 0.0066938002290 is a dimensionless 

coefficient 

 

Atmospheric Correction 

The mass of the atmosphere is unaccounted for in the theoretical gravity 

calculation and must be subtracted from the observed gravity.  The atmospheric 



correction uses the height h of the gravity station in meters above the GRS80 ellipsoid in 

the following equation (Hildebrand, 2002): 

Height Correction  

Measurements of observed gravity decrease with increasing distance from the 

center of the Earth.  In order to be compared with the theoretical gravity at the same 

location, the height of the gravity station must be corrected to the reference ellipsoid 

(Hildebrand, 2002): 

where, h is the height of the gravity station in meters above the GRS80 ellipsoid and φ is 

the latitude of the gravity station 

 

Bouguer Spherical Cap 

The Bouguer spherical cap correction sets the observed gravity value to a standard 

density based on either the average density of the continental crust (2.67 g/cm3) or to a 

site specific average density of the basement rock for local surveys.  Older methods of 

reducing gravity data used a similar correction called the Bouguer slab, which was based 

on a flat Earth model.  The Bouguer spherical cap correction is the new standard formula 

that accounts for the curvature of the Earth (Hildenbrand, 2002).   

where μ and λ are dimensionless coefficients (LaFehr, 1991), G is Newton’s gravitational 

constant = 6.6725985 x 10-11 m3 kg-1 s-2, ρ is the density of the spherical cap, usually 

2670 kg m-3, h is the height of the gravity station above the GRS80 reference ellipsoid 

(km), and R is the combined height of the gravity station and average radius of the Earth 

(km). 

 

GRAVITY SPREADSHEET 

The gravity spreadsheet calculates the corrections for instrument drift, height 

above the GRS80 reference ellipsoid, atmospheric, and the Bouguer spherical cap; as 

well as the DC shift for multiple day gravity surveys.  The meter-specific calibration 

table in the spreadsheet will convert gravimeter counter readings to corrected gravity 

measurements.  Tide and terrain corrections are not calculated in the spreadsheet, but 

users can enter values from other programs, such as InnerTC (Cogbill, 1990) in order to 

reduce gravity data to Bouguer anomalies.   

Prior to the standards set by the U.S. Geological Survey (Hildebrand, 2002), 

gravity reduction typically used orthometric heights (i.e., elevation with respect to mean 

sea level, or the geoid) to calculate Free-Air and Bouguer slab corrections.  In this 

spreadsheet, we conform to the USGS standards and employ ellipsoidal height 

corrections.  The revised method eliminates the need for the indirect effect, used to 



compensate for the difference between the ellipsoidal and geoidal heights in the Bouguer 

anomaly, as described by Hinze et al (2003) and Hildebrand (2002). 

 

Input Parameters 

 The four basic input parameters needed to calculate a Complete Bouguer anomaly 

are:  1) the height of the gravity station above the GRS80 reference ellipsoid, 2) the 

latitude of the station in WGS84 coordinates, 3) the drift and tide corrected observed 

gravity readings tied to an absolute gravity base station, and 4) the terrain correction for 

the location of the gravity station.  Additional utilities built into the spreadsheet are: 1) 

the instrument drift correction, 2) gravity meter dial conversions, 3) DC shift, and 4) 

conversion of local observed gravity to absolute gravity readings.  Refer to Appendix A 

for a detailed description of gravity spreadsheet use.  

 

Error Analysis 

 We referenced the output from the spreadsheet to the calculated Complete 

Bouguer anomalies determined by the USGS FORTRAN code used as the source for 

spreadsheet equations.  Using common values for gravity, altitude, and latitude, the 

calculations produced a difference of 2 to 14 μGals (i.e., 0.002 to 0.014 mGals).   

There are no systematic correlations among calculated gravity, altitude, and latitude (Fig. 

1), and the discrepancy probably arises from a truncation error within the spreadsheet 

algorithms.   



 

 Care must be used in studies that incorporate gravity values downloaded from the 

GeoNet server and those determined from the spreadsheet.  Although the same equations 

were used for both sets of calculations, the GeoNet server values were determined using 

orthometric versus ellipsoidal heights, which results in a discrepancy of as much as ± 7 

mGals.  This difference arises from the geoidal separation from the ellipsoid, which in 

North America can be as great as ± 20 m.   

 To reconcile this discrepancy, GeoNet values must be recalculated using 

ellipsoidal heights.  Given the latitude and longitude of each gravity station, the 

orthometric height can be transformed to the ellipsoidal height using the National 

Geodetic Survey (NGS) software program INTG.  INTG is a program developed by the 

United States government that uses the 2003 geoid model to calculate the geoidal 

separation (NGS, 2006).  The software is included with the gravity spreadsheet and can 

be downloaded from the NGS website (www.ngs.gov).   

 

CONCLUSIONS 

 The gravity spreadsheet is free and provides a simple tool for the reduction of raw 

gravity data to Bouguer anomalies, all in conformity with the standards set by the USGS 

and the North American Gravity Database Committee.  The spreadsheet eliminates the 

need for internet access by allowing the user to calculate the Bouguer anomaly of a 

gravity station in the field.  

 
APPENDIX A:   

GRAVITY SPREADSHEET v.1.0 INSTRUCTIONS 



The Gravity Spreadsheet v.1.0 is a Microsoft Excel workbook that is divided into 

several worksheets.  The primary worksheet entitled Gravity uses equations and values 

from four secondary worksheets (Bullard B Table, Meters, Calib. Table, Absolute Base) 

to calculate the Complete Bouguer anomaly.  The Bullard B Table worksheet is a list of 

constants used to calculate the Bouguer Spherical Cap correction (LaFehr, 1991) and 

should not be modified.  The Meters worksheet is a list of the three meters available for 

selection from the drop down menu in the spreadsheet and should not be modified.  The 

Calib. Table worksheet is a conversion table for counter values to gravity values for 

LaCoste-Romberg gravity meters and should be modified for the specific gravimeter 

used.  The Absolute Base worksheet references relative gravity measurements to an 

absolute value; the user must enter a drift and tide corrected measurement taken at the 

absolute base station.  Information for absolute base stations are available at PACES 

(http://paces.geo.utep.edu/). 

Each column in the Gravity worksheet is described below.  Included is a brief 

outline of the purpose of each column and the associated formula.   

 

Column A:  Gravity Station  

Gravity station identification.   

 

Column B:  Date  

Enter the date as MM/DD/YYYY.  Calculations for time durations use this format. 

 

Columns C and D:  Time – Hours/Minutes  



Enter hours in military time.  It is important use the 0-24 hour time scale because 

subsequent calculations depend on this format. 

 

Column E:  Duration (hours)  

This equation calculates the time, in decimal hours, that has elapsed from the initial base 

station reading.  These calculations are used to determine the drift correction. 

E3=((B3-$B$3)*24)+((C3+(D3/60))-($C$3+($D$3/60))) 

 

Columns F – K:  Latitude and Longitude  

Under both headings are three columns: d – degrees; m – minutes; sec- seconds.  If your 

coordinates are already in decimal degrees skip this data input section. 

 

Column L and M:  Latitude and Longitude (DD) 

DD stands for decimal degrees.  These columns convert coordinates that are in degrees, 

minutes and seconds into decimal degrees.   

L3=F3+(G3/60)+(H3/3600) 

M3=-1*(I3+(J3/60)+(K3/3600)) 

 

Column N:  Ellipsoid Height (m) 

If altitudes of gravity stations are in ellipsoidal height, select Ellipsoidal from the drop 

down menu and directly enter values into the column.  If the gravity station altitudes are 

in orthometric height, select Orthometric to Ellipsoidal from the drop down menu and 

supply values for Columns O and P (see below). 



 

Columns O and P:  Orthometric Height (m) and Geoid Hgt Separation (m) 

To convert orthometric to ellipsoidal heights use the NGS INTG.exe transformation 

software to calculate the geoidal separation.  Enter orthometric heights into Column O 

and enter the geoidal separation values into Column P.   

 

Columns Q and R:  Counter and Calibrated (mGal) 

These two columns are used to convert counter values to gravity for a LaCoste-Romberg 

gravimeter.  Each gravimeter has a calibration table that the user must manually input 

into the Calib. Table worksheet.  If the table for a specific gravimeter has more or fewer 

rows than that presented in the worksheet, then the cell range used in the equation under 

Column R must be modified.  For example, if the calibration sheet supplied with the 

LaCoste-Romberg gravity meter has one more row than the table in the Calib. Table 

worksheet then change the following equation in the Gravity worksheet from:  

 

R3 =(Q3-VLOOKUP(Q3,'Calib. Table'!$A$5:$C$75,1))*(VLOOKUP(Q3,'Calib. 

Table'!$A$5:$C$75,3))+VLOOKUP(Q3,'Calib. Table'!$A$5:$C$75,2) 

 

To: 

 

R3 =(Q3-VLOOKUP(Q3,'Calib. Table'!$A$5:$C$76,1))*(VLOOKUP(Q3,'Calib. 

Table'!$A$5:$C$76,3))+VLOOKUP(Q3,'Calib. Table'!$A$5:$C$76,2) 

 



This modifies cell R3 of the Gravity worksheet.  Drag the equation down for the 

remaining cells in the column.  (Hint: Select cell R3; place cursor over the small black 

box at the lower right hand corner of the cell; cursor will change to a black cross-hair; left 

click and drag the box down for the full range of cells) 

 

Column S:  Measured (mGals) 

For a digital gravity reading from a LaCoste-Romberg gravimeter, enter the values in this 

column (mGals). 

 

Column T:  Counter (Dial Reading) 

This column is for Worden gravity meter users.  The counter values from the dial 

readings are entered into this column.  

 

Column U:  Calibrated (mGals) 

In cell U1 enter the meter-specific constant for the Worden gravimeter.  The values 

entered in Column U are calibrated using this value. 

 

Column V:  Observed Gravity (mGals) 

Select the appropriate gravity meter from the drop down menu.  For Scintrex users, enter 

the observed gravity values recorded from the meter in Column V.  

 

Column W:  Tide (mgal) 



The user must make earth tide corrections using another program, because they are not 

calculated in this spreadsheet.  Enter the earth tide values for the specific time and 

location of each gravity measurement.   

 

Column X:  Tide Corrected (mgal) 

This column calculates the tide corrected gravity. 

X3 = V3- W3 

 

Column Y:  Meter Drift (mgal) 

The drift correction uses the base station gravity measurements at the beginning and end 

of the day, to calculate the rate of drift for the gravity measurements during the daily 

survey.  For surveys that involve multiple days, the drift correction is calculated for each 

day separately.  The following is given in the Example Spreadsheet:  

Day 1 (1/21/2006): 

Y3 =(($X$3-$X$12)/$E$12)*E3 

Day 2 (1/22/2006): 

Y13 = (($X$13-$X$22)/$E$22)*E3 

Day 3 (1/23/2006): 

Y23 = (($X$23-$X$34)/$E$34)*E3 

 

Column Z:  Drift & Tide Corrected (mGal) 

This column is the sum of Column X (Tide Corrected) and Column Y (Meter Drift) and 

produces a drift and tide corrected value of gravity. 



Z3 =X3+Y3 

 

Column AA:  DC Shift (mGals) 

This column supplies the values used in the DC Shift calculation in Column AG 

(Observed Absolute Gravity).  The DC shift is determined for the same relative base 

station among multiple day surveys and amongst multiple surveys.  The first occupation 

arbitrarily is used as the reference value.  Subsequent relative observations are 

differenced from the reference value after the values are drift and tide corrected.  The DC 

shift value is then entered manually for each individual day or suvey. 

 

Column AB:  Theoretical Gravity (mgals) 

The Somigliana closed-form formula (Hildenbrand, 2002) is used to calculate the 

theoretical gravity for each gravity station based on the GRS80 reference ellipsoid.  

AB3=100000*9.7803267714*((1+0.00193185138639*(SIN(L3*(PI()/180)))^2)/(S

QRT(1-0.00669437999013*(SIN(L3*(PI()/180)))^2))) 

 

Column AC:  Height Correction (mGals) 

This column corrects for the height of the gravity station relative to the GRS80 ellipsoid. 

AC3 =-0.308769097*N3+0.000439773125*N3*((SIN(L3*(PI()/180)))^2)+ 

0.0000000721251838*N3^2 

 

Column AD:  Atm Correction (mGals) 

This correction accounts for the mass of the atmosphere above the reference ellipsoid. 



AD3 = 0.874-0.000099*N3+0.00000000356*N3^2 

 

Column AE:  Bouguer Spherical Cap (mGals) 

The Bouguer Spherical Cap correction is the new standard method that accounts for the 

average mass and curvature of the Earth with respect to the ellipsoid.  This calculation 

uses the Bullard B Table.  The default density of the Bouguer spherical cap is 2.67 g/cc 

(LaFehr 1991). 

AE3 = (N3-VLOOKUP(N3,'Bullard B 

Table'!$A$4:$B$67,1))*((VLOOKUP(N3+100,'Bullard B Table'!$A$4:$B$67,2)-

VLOOKUP(N3,'Bullard B Table'!$A$4:$B$67,2))/100)+VLOOKUP(N3,'Bullard 

B Table'!$A$4:$B$67,2)+2*(PI())*(0.00000000006673)*N3*2670*100000 

 

Column AF:  Terrain Correction (mGals) 

Terrain corrections are not supported by the spreadsheet and need to be calculated using 

another software program.  Only enter the terrain correction value, not the sum of the 

observed gravity and terrain correction into Column AF.   

 

Column AG:  Observed Absolute Gravity (mGals) 

This column converts relative gravity observations to absolute gravity values.  The 

relative gravity value in Column Z (Drift and Tide Corrected) is referenced to an absolute 

gravity base station reading.  To complete this calculation, enter the absolute base station 

information in the Absolute Base worksheet. 

Absolute Base worksheet: 



Cell B6 is the absolute gravity value in mGals from the NGS information sheet. 

Cell B12 is the observed relative gravity value in mGals. 

Cell B13 is the tide correction in mGals. 

Cell B14 is the DC Shift in mGals. 

Cell B15 is the reading of the gravity meter after it has been corrected for tide and 

drift in mGals.  

 

AG3 =Z3-'Absolute Base'!$B$15+'Absolute Base'!$B$6+AA3 

 

Column AH:  Complete Bouguer Anomaly (mGals) 

The Complete Bouguer anomaly is the difference in observed gravity from the corrected 

theoretical gravity.   

AH3 =AG3-(AB3+AC3-AD3+AE3-AF3) 

Figure 1.  Scatter plot of the difference of Complete Bouguer anomalies between the gravity 

spreadsheet and the values calculated from USGS FORTRAN code. 

 


